Stanislav Trávníček

Oblouky

Rozhledy matematicko-fyzikální, Vol. 86 (2011), No. 4, 21–30

Persistent URL: http://dml.cz/dmlcz/146441

Oblouky

Stanislav Trávníček, PřF UP, Olomouc

Abstract. The paper shows how mathematical calculations and computational technology made it possible to solve a practical problem which occurred during a house construction. Computational technology enabled the user to choose the shape of the archs at the entrance terraces of the house. It also provided the information necessary for direct preparation of practical equipment needed for the construction to be realized.

Pro člověka 21. století, i když není informatik, se počítač stává pracovní pomůckou, která jemu nebo lidem blízkým pomáhá řešit různé problémy, s nimiž se setkává. A někdy k tomu ani nemusí používat nějaký speciální špičkový software a vystačí si s tím, co se naučil ve škole.

Projekt novostavby jednoho domu obsahoval kryté vstupní terasy, které byly shora uzavřeny vyzděnými oblouky. Tyto oblouky měly být všechny stejného charakteru, různé šířky, ale stejné výšky. Při vlastní stavbě zedníci (jen s malými zkušenostmi s oblouky) zhotovili pomocnou konstrukci, po jejímž osazení se ukázalo, že její tvar má do plánovaného tvaru kruhových oblouků docela daleko. Úkolem proto bylo nalézt jiný vhodný tvar oblouku, který by co nejvíce kopíroval tvar provedené stavební přípravy a přitom by oblouky zůstaly mírné a působily příjemně. Stanovení, který oblouk působí příjemně, je ovšem věcí subjektivního rozhodnutí, to nerozhodne počítač. Avšak pomocí počítače můžeme poskytnout podporu rozhodování tím, že předložíme k posouzení výběr různých tvarů oblouků; tak to v praxi skutečně proběhlo. Ukažme si nyní postup v daném konkrétním případě, při němž autor aktivně spolupracoval.

Pracovalo se za rozumného předpokladu, že všechny tyto oblouky mají být souměrné podle svislé osy, takže při jejich konstrukci stačí počítat jen jednu polovinu oblouku a pro znázornění druhé už pak lze využít osové souměrnosti. Zavedeme si souřadnicovou soustavu dle obr. 1 vzhledem k tomu, že číslování řádků v grafice jde shora dolů.

Šířku vstupu – velikost úsečky AB – nazveme $2G$, výšku oblouku AOB nazveme H; bod B má tedy souřadnice $[G; H]$ a bod A má souřadnice $[-G; H]$. Parametry G a H budou volitelné. Nyní uvážíme některé křivky, které by mohly sloužit jako horní oblouk vstupů do domu.



Obr. 1

a) Kruhový oblouk

Rovnice hledané kružnice (prochází počátkem a střed má na kladné poloose y) je

$$x^2 + (y - r)^2 = r^2,$$

tedy

$$x^2 + y^2 - 2ry = 0. \tag{1}$$

Dosaďme sem souřadnice bodu B a dostaneme

$$G^2 + H^2 - 2rH = 0,$$

odkud

$$r = \frac{G^2 + H^2}{2H}. \tag{2}$$

Z (1) pak máme

$$y = r - \sqrt{r^2 - x^2}, \tag{3}$$

kde r je dáno vztahem (2). (Před odmocninou je znaménko „$-$“, neboť nás zajímá „dolní“ oblouk kružnice.)

b) Oblouk kvadratické paraboly

Jde o parabolu s rovnicí

$$y = ax^2. \tag{4}$$

Dosaďme sem souřadnice bodu B, odkud $H = aG^2$, tedy

$$a = \frac{H}{G^2}.$$

Ze (4) pak máme

$$y = \frac{Hx^2}{G^2}. \tag{5}$$

c) Oblouk kubické paraboly

Jde o parabolu s rovnicí

$$y = ax^3.$$

Postupujeme stejně – dosadíme do rovnice souřadnice bodu B, odkud $H = aG^3$, tedy

$$a = \frac{H}{G^3}.$$

Dostaneme

$$y = \frac{Hx^3}{G^3}. \tag{6}$$

d) Oblouk elipsy

Rovnice elipsy (prochází počátkem a střed má na kladné poloose y) je

$$\frac{x^2}{a^2} + \frac{(y-b)^2}{b^2} = 1. \tag{7}$$

Zajímá nás „dolní“ půlelipsa, resp. její část, v okolí vedlejšího vrcholu O. Tuto část budeme volit pomocí reálného čísla $K \in (0, 1)$, a to tak, že $H = Kb$. (Pro $K = 1$ je obloukem celá půlelipsa, pro $K < 1$ jen její část kolem vedlejšího vrcholu.) Volbou H a K je tedy zadáno i b:

$$b = \frac{H}{K} \tag{8}$$

Do rovnice (7) dosaďme souřadnice bodu B:

$$\frac{G^2}{a^2} + \frac{(H-b)^2}{b^2} = 1$$

Druhý člen na levé straně je

$$\frac{(H-b)^2}{b^2} = \left(h - \frac{H}{K}\right)^2 \cdot \frac{K^2}{H^2} = K^2 - 2K + 1,$$

takže

$$a^2 = \frac{G^2}{K(2-K)}.$$

Ze (7) máme

$$(y-b)^2 = b^2\left(1-\frac{x^2}{a^2}\right),$$

odkud

$$y = \frac{H}{K}\left(1-\sqrt{1-\frac{K(2-K)x^2}{G^2}}\right). \qquad (9)$$

Nyní můžeme na základě zadání G, H, K tyto křivky vykreslovat. Pro větší názornost přidejme ještě svislé boční stěny (od bodů A, B dolů) o délce $H0$. Zadává se tedy: výška boků, tj. $H0$, např. 200 cm, výška oblouku, tj. H, např. 30 cm, poloviční šířka, tj. G, např. 100 cm a při volbě elipsy se zadává ještě koeficient K, např. 0,7.

Sestavíme-li program jen se zcela minimálním komfortem, ale tak, aby byl názorný pro uživatele, můžeme dostat např. tento:

```
program Oblouky;
{$F+}
uses Graph, Crt;
const
  Ox = 100;  {poloha osy x: pocet bodu shora}
  Oy = 320;  {poloha osy y: pocet bodu zleva}
type  UserFunc = function(X: Real): Real;
var
  grDriver, grMode, ErrCode: Integer;
  OrigMode, DruhObl, G, H0, H: Integer;
  K: Real;

function Y1(X: Real): Real;     {kruznice (3)}
var Q, R: Real;
begin {Y1}
  R := (G * G + H * H)/(2 * H);
  Q := (R * R - X * X);
  Y1 := R - Sqrt(Q)
end;  {Y1}

function Y2(X: Real) : Real;    {kvadr.parabola (5)}
begin {Y2}
  Y2 := H * (X / G) * (X / G)
end; {Y2}
```

```
function Y3(X: Real): Real;     {kubic.parabola (6)}
begin  {Y3}
  Y3 := H * (X / G) * (X / G) * (X / G)
end; {Y3}

function Y4(X: Real) : Real;    {cast K pulelipsy (9)}
var Q: Real;
begin  {Y4}
  Q := 1 - K * (2 - K) * (X / G) * (X / G);
  Y4 := (H / K) * (1 - Sqrt(Q))
end; {Y4}

procedure SestrojBoky;
var Ii,Jj : Integer;
begin SestrojBoky
  SetColor(White);
  Line(-G+Oy,Ox+H,-G+Oy, Ox+HO+H);
  Line(-G+Oy-1,Ox+H,-G+Oy-1, Ox+HO+H);
  Line(G+Oy,Ox+H,G+Oy, Ox+HO+H);
  Line(G+Oy+1,Ox+H,G+Oy+1, Ox+HO+H)
end;  {SestrojBoky}

procedure ZobrazFunkci(F: UserFunc; FuncColor: Integer);
var
  Xx, Yy: LongInt; {souradnice na obrazovce}
  X, Y: Real;    {skutecne hodnoty}
begin {ZobrazFunkci}
  for Xx := 0 to G do
  begin
    X := Xx;
    Y := F(X);
    Yy := Round(Y) + Ox;
    PutPixel(Xx + Oy, Yy, FuncColor);
    PutPixel(Xx + Oy, Yy-1, FuncColor);
    PutPixel(-Xx+ Oy, Yy, FuncColor);
    PutPixel(-Xx+ Oy, Yy-1, FuncColor)
  end
end;  {ZobrazFunkci}
```

```
begin program
  repeat
  OrigMode := LastMode; TextMode(C80+Font8x8);
  ClrScr;
  WriteLn('Oblouky:');
  WriteLn;
  WriteLn('1 = kruhovy');
  WriteLn('2 = kvadraticka parabola');
  WriteLn('3 = kubicka parabola');
  WriteLn('4 = elipticky');
  WriteLn('0 = konec');
  WriteLn; Write('Volba: ');
  ReadLn(DruhObl);
  if not(DruhObl in [1..4]) then Halt;
  WriteLn;
  WriteLn('Zadej v cm:');
  Write('vyska boku = '); ReadLn(H0);
  Write('vyska oblouku = '); ReadLn(H);
  Write('polovicni sirka = '); ReadLn(G);
  if DruhObl = 4 then
  begin
    Write('Zadej K v (0,1): '); ReadLn(K)
  end;
{Zobrazeni}
  grDriver := Detect;
  InitGraph(grDriver,grMode,'');
  ErrCode := GraphResult;
  if ErrCode = grOk then
  begin
    SestrojBoky;
    case DruhObl of
    1: ZobrazFunkci(Y1,Yellow);
    2: ZobrazFunkci(Y2,Yellow);
    3: ZobrazFunkci(Y3,Yellow);
    4: ZobrazFunkci(Y4,Yellow)
    end;
    ReadLn; CloseGraph
  end else
```

```
  begin
    CloseGraph;
    WriteLn('Graphics error: ', GraphErrorMsg(ErrCode))
  end
  until false
end.  {program}
```

Užitím tohoto programu je možné (vhodnou volbou parametrů) předvádět různé případy oblouků. Uživatel zadá výšku a šířku svých vstupů i výšku a tvar oblouků a na základě jejich zobrazení v příjemné velikosti posoudí jejich estetické působení. Program volí měřítko 1 cm = 1 bod obrazovky, takže není třeba nic přepočítávat, čáry jsou zdvojeny, takže na obrazovce je zajištěna i zřetelnost a názornost. Ukázky výsledků práce programu jsou na obr. 2a až 2d. V obrázcích je zvoleno $G = 75$ cm, $H = 30$ cm, $H0 = 50$ cm (zde jen pro šetření místem v časopisu, ale celkový dojem je možné lépe posoudit, když se zadá výška H0 bočních zdí v plné hodnotě).



Obr. 2a: Kruhový oblouk



Obr. 2b: Kvadratická parabola

Obr. 2c: Kubická parabola



Obr. 2d: Eliptický oblouk, $K = 0{,}7$

Oblouků elipsy lze vytvořit více nejen volbou různých výšek a šířek oblouku, ale též různou volbou parametru K. V daném konkrétním případě se posuzují jednotlivé oblouky pro všechny projektované vstupní terasy o zadaných výškách a šířkách a po ukázce více případů může zvítězit u stavebníka eliptický oblouk s parametrem $K = 0{,}5$ (obr. 3; zde $G = 125$ cm, $H = 40$ cm).



Obr. 3: Eliptický oblouk, $K = 0{,}5$

Reálně zadané parametry vedou k vytvoření požadovaných oblouků (obr. 4a až 4c).

Obr. 4a: Čelní vstupní terasa

Obr. 4b: Zadní terasy

Obr. 4c: Boční vstupy

Pomoc při rozhodování o tvaru oblouků však je jen první částí akce. Počítač pak totiž pomůže i při samotné stavbě domu. V Excelu se dají vytvořit tabulky souřadnic bodů vybraných oblouků a pomocí nich zhotovit šablony určené přímo pro zedníky vytvářející a vyzdívající zvolené oblouky (obr. 5). Připomeňme, že x jsou vzdálenosti od vrcholu oblouku k jeho okrajům.

	A	B	C	D	E	F	G	H	I	J	K
1	Eliptický oblouk		G =	1,00	H =	0,30	K =	0,50			
2											
3	x =	Q =	y =	0,1+y							
4	0,00	1	0,000	0,100							
5	0,10	0,9925	0,002	0,102							
6	0,20	0,97	0,009	0,109							
7	0,30	0,9325	0,021	0,121							
8	0,40	0,88	0,037	0,137							
9	0,50	0,8125	0,059	0,159							
10	0,60	0,73	0,087	0,187							
11	0,70	0,6325	0,123	0,223							
12	0,80	0,52	0,167	0,267							
13	0,90	0,3925	0,224	0,324							
14	1,00	0,25	0,300	0,400							



Obr. 5

Zde je proveden výpočet pro oblouk šířky $2G = 2$ m, ale vzhledem k jeho symetrii se uvažuje jen jeho polovina. Výška oblouku je zde $H = 30$ cm, koeficient $K = 0{,}5$. Výraz pod odmocninou ve vzorci (9) dostaneme jako Q ve sloupci B

=1-H1*(2−H1)*A4*A4/(D1*D1)

(příkaz umístíme do B4) a odsud vypočteme y do sloupce C vztahem

=\$F\$1*(1−ODMOCNINA(B4))/\$H\$1

(příkaz umístíme do C4). To slouží pro vytvoření šablon, tedy lišt, na nichž je odměřeno $0{,}1 + y$ dolů od vodorovné linky vedené 10 cm nad vrcholem oblouku (obr. 6a,b).

Obr. 6a: Vodorovná linka

Obr. 6b: Definitivní vyzdívání

Obr. 6c: Hotovo